Fall 1999 Math 151 Common Exam 2 Solutions

Sections 3.3-4.2

[Maple V Release 5.1 worksheet]

```
> restart;
Warning, new definition for norm
Warning, new definition for trace
Package:   vec_calc   Version 4.3
For all HELP, execute: ?vec_calc
To use aliases, execute:   vc_aliases;
```

A1 / B2

```
> exp(x^3 + x); derivative:=diff(%, x);
```

$$\mathbf{e}^{(x^3+x)}$$

$$derivative := (3\,x^2+1)\,\mathbf{e}^{(x^3+x)}$$

A2 / B3

```
> f:=x->sin(x)^2; derivative:=D(f)(Pi/4);
```

$$f := x \rightarrow \sin(x)^2$$

$$derivative := 1$$

A3 / B5

```
> sin(sqrt(x^2 + 1)); derivative:=diff(%, x);
```

$$\sin(\sqrt{x^2+1}\,)$$

$$derivative := \frac{\cos(\sqrt{x^2+1}\,)\,x}{\sqrt{x^2+1}}$$

A4 / B6

```
> f:=x->x^(1/3);
  f(8)  +  D(f)(8) * (x - 8);
  L:=unapply(%, x); approximation:=simplify(L(8.012));
```

$$f := x \rightarrow x^{(1/3)}$$

$$8^{(1/3)} + \frac{1}{24}\,8^{(1/3)}\,(x-8)$$

$$L := x \to 8^{(1/3)} + \frac{1}{24} 8^{(1/3)} (x - 8)$$

$$approximation := 2.001000000$$

A5 / B1

```
> simplify(3^x + 3^x + 3^x);
```

$$3^{(1+x)}$$

A6 / B4

```
> x*exp(x); diff(%, x$20);
```

$$x \, \mathbf{e}^x$$

$$20 \, \mathbf{e}^x + x \, \mathbf{e}^x$$

A7 / B12

```
> sin(Pi/3) = h/x; op(solve(%, {x}));
  subs(%, P=P(t), h=h(t), P = 3*x);
  diff(%, t); subs(diff(h(t), t)=2, %);
```

$$\frac{1}{2}\sqrt{3} = \frac{h}{x}$$

$$x = \frac{2}{3} h \sqrt{3}$$

$$\mathrm{P}(t) = 2 \, \mathrm{h}(t) \sqrt{3}$$

$$\frac{\partial}{\partial t} \mathrm{P}(t) = 2 \left(\frac{\partial}{\partial t} \mathrm{h}(t) \right) \sqrt{3}$$

$$\frac{\partial}{\partial t} \mathrm{P}(t) = 4 \sqrt{3}$$

A8 / B7

```
> unassign('f'); x[2] = x[1] - f(x[1]) / Df(x[1]);
  subs(Df(x[1]) = m, f(x[1])=3/2, x[2]=3/5, x[1]=1, %);
  op(solve(%, {m}));
```

$$x_2 = x_1 - \frac{\mathrm{f}(x_1)}{\mathrm{Df}(x_1)}$$

$$\frac{3}{5} = 1 - \frac{3}{2} \frac{1}{m}$$

$$m = \frac{15}{4}$$

A9 / B10

```
> Limit(sin(8*x)^2 / (3*x)^2, x=0); value(%);
```

$$\lim_{x \to 0} \frac{1}{9} \frac{\sin(8\,x)^2}{x^2}$$

$$\frac{64}{9}$$

A10 / B8

```
> x:=t->t^2; y:=t->t^3-3*t; r:=t->[x(t),y(t)];
  t_vals:=solve(D(y)(t)=0, t);
  check_dx_dt_nonzero:=D(x)(1)<>0, D(x)(-1)<>0;
  xy_pairs:=r(1), r(-1);
>
```

$x := t \to t^2$

$y := t \to t^3 - 3\,t$

$r := t \to [\mathrm{x}(t), \mathrm{y}(t)]$

$t\_vals := 1, -1$

$check\_dx\_dt\_nonzero := 2 \neq 0, -2 \neq 0$

$xy\_pairs := [1, -2], [1, 2]$

A11 / B9

```
> unassign('L', 'x', 'y');
  f:=x->sqrt(1+x);
  L(x) = f(0)  +  D(f)(0) * (x - 0);
```

$f := x \to \sqrt{x + 1}$

$$\mathrm{L}(x) = 1 + \frac{1}{2}\,x$$

A12 / B11

```
> unassign('f'); Dg(x) = 1/Df(g(x));
  subs(x=2, %); subs(g(2)=4, %); subs(Df(4)=3, %);
```

$$\mathrm{Dg}(x) = \frac{1}{\mathrm{Df}(\mathrm{g}(x))}$$

$$\mathrm{Dg}(2) = \frac{1}{\mathrm{Df}(\mathrm{g}(2))}$$

$$\mathrm{Dg}(2) = \frac{1}{\mathrm{Df}(4)}$$

$$\mathrm{Dg}(2) = \frac{1}{3}$$

A13 / B16

```
> z(t)^2 = y(t)^2 + 50^2;
  diff(%, t); %/(2*z(t));
  lhs(%) = subs(diff(y(t), t)=44, y(t)=120, z(t)=sqrt(120^2 +
  50^2), rhs(%));
  evalf(%), ft/s;
```

$$\mathrm{z}(t)^2 = \mathrm{y}(t)^2 + 2500$$

$$2\,\mathrm{z}(t)\left(\frac{\partial}{\partial t}\mathrm{z}(t)\right) = 2\,\mathrm{y}(t)\left(\frac{\partial}{\partial t}\mathrm{y}(t)\right)$$

$$\frac{\partial}{\partial t}\mathrm{z}(t) = \frac{\mathrm{y}(t)\left(\frac{\partial}{\partial t}\mathrm{y}(t)\right)}{\mathrm{z}(t)}$$

$$\frac{\partial}{\partial t}\mathrm{z}(t) = \frac{528}{13}$$

$$\frac{\partial}{\partial t}\mathrm{z}(t) = 40.61538462, \frac{ft}{s}$$

A14 / B14

(a)

```
> g:=x->f(x)^3;
  derivative_in_general:=D(g)(x);
  derivative_in_particular:=D(g)(3);
  answer:=subs(D(f)(3)=4, f(3)=6, %);
```

$$g := x \rightarrow \mathrm{f}(x)^3$$

$$derivative\_in\_general := 3\,\mathrm{f}(x)^2\,\mathrm{D}(f)(x)$$

$$derivative\_in\_particular := 3\,\mathrm{f}(3)^2\,\mathrm{D}(f)(3)$$

$$answer := 432$$

(b)

```
> h:=x->f(f(x));
  derivative_in_general:=D(h)(x);
  derivative_in_particular:=D(h)(3);
  subs(D(f)(3)=4, f(3)=6, %);
  answer:=subs(D(f)(6)=-1, %);
```

$h := x \to f(f(x))$

$derivative\_in\_general := D(f)(f(x))\, D(f)(x)$

$derivative\_in\_particular := D(f)(f(3))\, D(f)(3)$

$4\, D(f)(6)$

$answer := -4$

A15 / B17

```
> f:=x->x / sqrt(2-x^2);
  Use_the_point_slope_formula:=y - f(1) = D(f)(1) * (x - 1);
  tangent_line:=% + (1=1);
```

$$f := x \to \frac{x}{\sqrt{2 - x^2}}$$

$Use\_the\_point\_slope\_formula := y - 1 = 2\, x - 2$

$tangent\_line := y = 2\, x - 1$

A16 / B13

Since the slopes of the curves at the point of intersection are negative reciprocals, the tangent lines to the curves thereat are perpendicular.

```
> eq1:=x^2 + y^2 = 4*x; eq2:=x^2 + y^2 = 2*y;
  m1:=implicitdiff(eq1, y, x); m2:=implicitdiff(eq2, y, x);
  slope_1_at_intersection:=subs(x=4/5, y=8/5, m1);
  slope_2_at_intersection:=subs(x=4/5, y=8/5, m2);
```

$eq1 := x^2 + y^2 = 4\, x$

$eq2 := x^2 + y^2 = 2\, y$

$$m1 := -\frac{x - 2}{y}$$

$$m2 := -\frac{x}{y - 1}$$

$$slope\_1\_at\_intersection := \frac{3}{4}$$

$$slope\_2\_at\_intersection := \frac{-4}{3}$$

A17 / B15

```
> r:=MF(t, [6*sin(t), 3*cos(t)]);
  v:=D(r); a:=D(v); s:=MF(t, len(v(t)));
```

$r := [t \to 6\sin(t), t \to 3\cos(t)]$

$v := [t \to 6\cos(t), t \to -3\sin(t)]$

$a := [t \to -6\sin(t), t \to -3\cos(t)]$

$s := t \to 3\sqrt{4\cos(t)^2 + \sin(t)^2}$

(a)

```
> r(Pi/4); evalf(%);
```

$$\left[3\sqrt{2}, \frac{3}{2}\sqrt{2}\right]$$

[4.242640686, 2.121320343]

(b)

```
> v(Pi/4); evalf(%);
```

$$\left[3\sqrt{2}, -\frac{3}{2}\sqrt{2}\right]$$

[4.242640686, -2.121320343]

(c)

```
> combine(s(Pi/4)); evalf(%);
```

$$\frac{3}{2}\sqrt{10}$$

4.743416490

(d)

```
> a(Pi/4); evalf(%);
```

$$\left[-3\sqrt{2}, -\frac{3}{2}\sqrt{2}\right]$$

[-4.242640686, -2.121320343]

(e)

```
> with(plottools):
  plot_of_ellipse:=plot([op(r(t)), t=0..2*Pi],
    scaling=constrained, thickness=3):
  P:=r(Pi/4): Q:=P + v(Pi/4): R:=P + a(Pi/4):
  plot_of_velocity:=arrow(P, Q, 0.1, 0.4, 0.1, color=green):
  plot_of_acceleration:=arrow(P, R, 0.1, 0.4, 0.1, color=blue):
```

```
tv:=textplot([7,1.7,‘velo‘],align={ABOVE,CENTER}):
ta:=textplot([2,1.7,‘acc‘],align={ABOVE,CENTER}):
display([plot_of_ellipse, plot_of_velocity,
  plot_of_acceleration, tv, ta]);
```



Bonus

```
> unassign('f', 'g', 'x');
  g:=x->f(x^2 - x); D(g)(x) = 0;
  eq1:=x^2 - x = 2; eq2:=x^2 - x = 6; eq3:=2*x - 1 = 0;
  solve(eq1, {x}); solve(eq2, {x}); solve(eq3, {x});
```

$g := x \to \mathrm{f}(x^2 - x)$

$\mathrm{D}(f)(x^2 - x)\,(2\,x - 1) = 0$

$eq1 := x^2 - x = 2$

$eq2 := x^2 - x = 6$

$eq3 := 2\,x - 1 = 0$

$\{x = -1\}, \{x = 2\}$

$\{x = -2\}, \{x = 3\}$

$\{x = \frac{1}{2}\}$